

Principles of Linear Algebra With *Maple*<sup>TM</sup>  
Sections 10.4 and 10.5 *Revised*

Kenneth Shiskowski and Karl Frinkle

© *Revision date November 2, 2010*



## Chapter 10

# The Geometry of Linear and Affine Maps

### 10.4 Rotations, Reflections, and Rescalings in Three Dimensions

In this section, we look at those linear maps  $T : \mathbb{R}^3 \rightarrow \mathbb{R}^3$  that are rotations about lines through the origin, reflections about a coordinate axis or the origin, and general rescalings. Let us begin by discussing rotations in  $\mathbb{R}^3$ , in particular we already know how to rotate objects by an angle  $\theta$  in the  $xy$ -plane. In three dimensions, a rotation in the  $xy$ -plane means a rotation about the  $z$ -axis. Therefore, we need to construct a  $3 \times 3$  matrix that keeps all  $z$  values the same, and rotates the  $(x, y)$  coordinates by an angle of  $\theta$ . Your first guess should be to put the  $2 \times 2$  matrix in the upper righthand corner of our hopeful matrix. The question remains, What do we do about the last row and column? Well, we wish for  $z$  to stay fixed, therefore we should use the third row and column of the identity matrix. So our matrix looks as follows:

$$A_{\theta_z} = \begin{bmatrix} \cos(\theta) & -\sin(\theta) & 0 \\ \sin(\theta) & \cos(\theta) & 0 \\ 0 & 0 & 1 \end{bmatrix} \quad (10.26)$$

which we can verify works if we perform the multiplication  $A_{\theta_z} \vec{v}$ , where  $\vec{v} \in \mathbb{R}^3$ :

$$\begin{bmatrix} \cos(\theta) & -\sin(\theta) & 0 \\ \sin(\theta) & \cos(\theta) & 0 \\ 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} x \\ y \\ z \end{bmatrix} = \begin{bmatrix} x \cos(\theta) - y \sin(\theta) \\ x \sin(\theta) + y \cos(\theta) \\ z \end{bmatrix}$$

The matrix  $A_{\theta_z}$  can be easily modified to get the two other axis rotation matrices:

$$A_{\theta_x} = \begin{bmatrix} 1 & 0 & 0 \\ 0 & \cos(\theta) & -\sin(\theta) \\ 0 & \sin(\theta) & \cos(\theta) \end{bmatrix}, \quad A_{\theta_y} = \begin{bmatrix} \cos(\theta) & 0 & -\sin(\theta) \\ 0 & 1 & 0 \\ \sin(\theta) & 0 & \cos(\theta) \end{bmatrix} \quad (10.27)$$

Now we want to find the matrix  $A_{\vec{v}}$  that allows us to rotate about any line through the origin in three-space that is parallel to the unit vector  $\vec{v}$ , where no component of  $\vec{v}$  is zero. If any component of  $\vec{v}$  is zero, then we would simply be rotating about one of the axes. If you analyze what we did in Section ??, then you find that the rotation matrix  $A_{\vec{v}}$  has columns gotten by revolving respectively each column of  $I_3$  from the  $3 \times 3$  identity matrix.

```
> with(linalg): with(plots): with(plottools):
> assume(alpha, real); assume(beta, real); assume(delta, real);
> interface(showassumed = 0):
> v:= [alpha,beta,delta]:
> Q1:= [1,0,0]:
> C1:= evalm(dotprod(v,Q)*v);
```

$$C := \begin{bmatrix} \alpha^2 & \alpha\beta & \alpha\delta \end{bmatrix}$$

```
> R1:= sqrt(1-alpha^2);
```

$$R1 := \sqrt{1 - \alpha^2}$$

```
> U1:= evalm((Q1-C1)/R1);
```

$$U1 := \begin{bmatrix} \sqrt{1 - \alpha^2} & -\frac{\alpha\beta}{\sqrt{1 - \alpha^2}} & -\frac{\alpha\delta}{\sqrt{1 - \alpha^2}} \end{bmatrix}$$

```
> V1:= simplify(crossprod(U1,v), {alpha^2+beta^2+delta^2=1});
```

$$V1 := \begin{bmatrix} 0 & -\frac{\delta}{\sqrt{1 - \alpha^2}} & \frac{\beta}{\sqrt{1 - \alpha^2}} \end{bmatrix}$$

```
> Nv1:= expand(simplify(evalm(C1+R1*cos(theta)*U1+R1*sin(theta)*V1)));
```

$$Nv1 := [\alpha^2 + \cos(\theta) - \cos(\theta)\alpha^2 \quad \alpha\beta - \cos(\theta)\alpha\beta - \sin(\theta)\delta \quad \alpha\delta - \cos(\theta)\alpha\delta + \sin(\theta)\beta]$$

This is the first column of the rotation matrix  $A_{\vec{v}}$ . Now let us find its second and third columns.

```
> Q2:= [0,1,0]:
```

> C2:= evalm(dotprod(v,Q2)\*v);

$$C2 := \begin{bmatrix} \alpha\beta & \beta^2 & \beta\delta \end{bmatrix}$$

> R2:= sqrt(1-beta^2);

$$R2 := \sqrt{1-\beta^2}$$

> U2:= evalm((Q2-C2)/R2);

$$U2 := \begin{bmatrix} -\frac{\alpha\beta}{\sqrt{1-\beta^2}} & \sqrt{1-\beta^2} & -\frac{\beta\delta}{\sqrt{1-\beta^2}} \end{bmatrix}$$

> V2:= simplify(crossprod(U2,v),{alpha^2+beta^2+delta^2=1});

$$V2 := \begin{bmatrix} \frac{\delta}{\sqrt{1-\beta^2}} & 0 & -\frac{\alpha}{\sqrt{1-\beta^2}} \end{bmatrix}$$

> Nv2:= expand(simplify(evalm(C2+R2\*cos(theta)\*U2+R2\*sin(theta)\*V2)));

$$Nv2 := [\alpha\beta - \cos(\theta)\alpha\beta + \sin(\theta)\delta \beta^2 + \cos(\theta) - \cos(\theta)\beta^2 \beta\delta - \cos(\theta)\beta\delta - \sin(\theta)\alpha]$$

> Q3:= [0,0,1]:

> C3:= evalm(dotprod(v,Q3)\*v);

$$C3 := \begin{bmatrix} \alpha\delta & \beta\delta & \delta^2 \end{bmatrix}$$

> R3:= sqrt(1-delta^2);

$$R := \sqrt{1-\delta^2}$$

> U3:= evalm((Q3-C3)/R3);

$$U3 := \begin{bmatrix} -\frac{\alpha\delta}{\sqrt{1-\delta^2}} & -\frac{\beta\delta}{\sqrt{1-\delta^2}} & \sqrt{1-\delta^2} \end{bmatrix}$$

> V3:= simplify(crossprod(U3,v),{alpha^2+beta^2=1-delta^2});

$$V3 := \begin{bmatrix} -\frac{\beta}{\sqrt{1-\delta^2}} & \frac{\alpha}{\sqrt{1-\delta^2}} & 0 \end{bmatrix}$$

> Nv3:= expand(simplify(evalm(C3+R3\*cos(theta)\*U3+R3\*sin(theta)\*V3)));

$$Nv3 := [\alpha\delta - \cos(\theta)\alpha\delta - \sin(\theta)\beta \beta\delta - \cos(\theta)\beta\delta + \sin(\theta)\alpha \delta^2 + \cos(\theta) - \cos(\theta)\delta^2]$$

Next, we have all three columns of  $A_{\vec{v}}$ . Notice that no component of the unit vector  $\vec{v} = \langle \alpha, \beta, \delta \rangle$  parallel to our axis of rotation can be zero since we must divide by their absolute values in  $A_{\vec{v}}$ .

> rotmatrix:= transpose(matrix([Nv, Nv2, Nv3]));

rotmatrix :=

$$\begin{bmatrix} \alpha^2 + \cos(\theta) - \cos(\theta)\alpha^2 & \alpha\beta - \cos(\theta)\alpha\beta - \sin(\theta)\delta & \alpha\delta - \cos(\theta)\alpha\delta + \sin(\theta)\beta \\ \alpha\beta - \cos(\theta)\alpha\beta + \sin(\theta)\delta & \beta^2 + \cos(\theta) - \cos(\theta)\beta^2 & \beta\delta - \cos(\theta)\beta\delta - \sin(\theta)\alpha \\ \alpha\delta - \cos(\theta)\alpha\delta - \sin(\theta)\beta & \beta\delta - \cos(\theta)\beta\delta + \sin(\theta)\alpha & \delta^2 + \cos(\theta) - \cos(\theta)\delta^2 \end{bmatrix}$$

> A:= (alpha,beta,delta,theta) -> matrix([[alpha^2 + cos(theta) - cos(theta)\*alpha^2, alpha\*beta - cos(theta)\*alpha\*beta + sin(theta)\*delta, alpha\*delta - cos(theta)\*alpha\*delta - sin(theta)\*beta], [alpha\*beta - cos(theta)\*alpha\*beta - sin(theta)\*delta, beta^2 + cos(theta) - cos(theta)\*beta^2, beta\*delta - cos(theta)\*beta\*delta + sin(theta)\*alpha], [alpha\*delta - cos(theta)\*alpha\*delta + sin(theta)\*beta, beta\*delta - cos(theta)\*beta\*delta - sin(theta)\*alpha, delta^2 + cos(theta) - cos(theta)\*delta^2]]):

> evalf(A(2/sqrt(38),-5/sqrt(38),3/sqrt(38),9\*Pi/7));

$$\begin{bmatrix} -0.4525961384 & -0.8077236010 & 0.8904895659 \\ -0.8077236010 & 0.4445955941 & -0.3871916093 \\ -0.3778085756 & -0.3871916092 & -0.2389790592 \end{bmatrix}$$

The matrix above is an example with  $\vec{v} = \frac{1}{\sqrt{38}}\langle 2, -5, 3 \rangle$  and  $\theta = \frac{9}{7}\pi$ . The above calculations may seem lengthy, tedious, and complicated, however, if we spend a few minutes working through the details, perhaps we can demystify this whole process. Remember that  $I_3$  is the  $3 \times 3$  matrix whose columns are the standard basis vectors  $\{\vec{e}_1, \vec{e}_2, \vec{e}_3\}$  of  $\mathbb{R}^3$ . We will rotate  $\vec{e}_k$  about  $\vec{v} = \langle \alpha, \beta, \delta \rangle$  by an angle of  $\theta$ , where  $k = 1, 2, 3$  and  $\alpha^2 + \beta^2 + \delta^2 = 1$ . Once again following the pattern of Section 7.1, we must find the point  $C$  (or in vector form  $\vec{c}$ ) on  $\vec{v}$  closest to the tip of vector  $\vec{e}_k$ . In vector form, this is simply  $\text{proj}_{\vec{v}}(\vec{e}_k)$ :

$$\vec{c} = \frac{\vec{v} \cdot \vec{e}_k}{|\vec{v}|} \vec{v} \quad (10.28)$$

Under the assumption that  $|\vec{v}| = 1$ , we find that  $\vec{c} = (\vec{v} \cdot \vec{e}_k) \vec{v}$ . Now that we have the center of rotation, we need to construct our two coordinate axes to rotate about. These coordinate axes will be the plane perpendicular to  $\vec{v}$  centered at the point  $C$ . To find the first axis, notice that  $\vec{e}_k - \vec{c}$  is perpendicular to  $\vec{c}$ , with  $R = |\vec{e}_k - \vec{c}|$  being the distance from the tip of vector  $\vec{e}_k$  to the point  $C$ . So we define

$$\vec{u} = \frac{1}{R} (\vec{e}_k - \vec{c}) \quad (10.29)$$

to be the unit vector orthogonal to  $\vec{c}$  that starts at the point  $C$  and points in the direction of  $\vec{e}_k$ . A unit vector orthogonal to both  $\vec{c}$  and  $\vec{u}$  is easy to

compute by using the cross product:

$$\begin{aligned}\vec{w} &= \frac{\vec{u} \times \vec{c}}{|\vec{u} \times \vec{c}|} \\ &= \vec{u} \times \vec{v}\end{aligned}\tag{10.30}$$

The above two definitions of  $\vec{w}$  are equivalent since  $\vec{c}$  and  $\vec{v}$  are in the same direction and both cross products result in a unit length vector. To see this, first we note that the first is obviously unit length; secondly, since  $\vec{u}$  and  $\vec{v}$  are unit length and perpendicular, we are guaranteed that  $\vec{u} \times \vec{v}$  will be of unit length by the formula since

$$|\vec{u} \times \vec{v}| = |\vec{u}| |\vec{v}| \sin\left(\frac{\pi}{2}\right)$$

From a computational point of view, it is easier to compute  $\vec{u} \times \vec{v}$ , however for matters of simplification, which we need soon, we choose the first definition. Regardless of which formula you use for  $\vec{w}$ , the set  $\{\vec{u}, \vec{w}\}$  forms an orthonormal basis for the plane perpendicular to  $\vec{v}$ . So to rotate  $\vec{e}_k$  by an angle of  $\theta$  about  $\vec{v}$ , which we will denote as  $\vec{e}_k(\theta_{\vec{v}})$ , we simply perform the following vector operation:

$$\vec{e}_k(\theta_{\vec{v}}) = \vec{c} + R \cos(\theta) \vec{u} + R \sin(\theta) \vec{w}\tag{10.31}$$

As it stands, this formula may look simple, however,  $\vec{u}$  and  $\vec{w}$  both are more complicated expressions. Let us investigate this further. By looking at the definition of  $\vec{u}$ , we see that

$$\begin{aligned}R \cos(\theta) \vec{u} &= R \cos(\theta) \frac{\vec{e}_k - \vec{c}}{R} \\ &= \cos(\theta) (\vec{e}_k - \vec{c})\end{aligned}\tag{10.32}$$

The final term requires a little more work, but it does simplify after some algebraic maneuvers:

$$\begin{aligned}R \sin(\theta) \vec{w} &= R \sin(\theta) \frac{\vec{c} \times \frac{\vec{e}_k - \vec{c}}{R}}{\left| \vec{c} \times \frac{\vec{e}_k - \vec{c}}{R} \right|} \\ &= R \sin(\theta) \frac{\frac{1}{R} (\vec{c} \times (\vec{e}_k - \vec{c}))}{\frac{1}{R} |\vec{c} \times (\vec{e}_k - \vec{c})|} \\ &= R \sin(\theta) \frac{(\vec{c} \times (\vec{e}_k - \vec{c}))}{|\vec{c} \times (\vec{e}_k - \vec{c})|}\end{aligned}\tag{10.33}$$

Now, we also have that

$$\begin{aligned} |\vec{c} \times (\vec{e}_k - \vec{c})| &= |\vec{c}| |\vec{e}_k - \vec{c}| \sin(\alpha) \\ &= |\vec{c}| R \sin\left(\frac{\pi}{2}\right) \\ &= |\vec{c}| R \end{aligned} \quad (10.34)$$

and

$$\begin{aligned} \vec{c} \times (\vec{e}_k - \vec{c}) &= \vec{c} \times \vec{e}_k - \vec{c} \times \vec{c} \\ &= \vec{c} \times \vec{e}_k \end{aligned} \quad (10.35)$$

Putting equations (10.34) and (10.35) into (10.33) gives

$$R \sin(\theta) \vec{w} = \frac{\vec{c} \times \vec{e}_k}{|\vec{c}|} \sin(\theta) \quad (10.36)$$

After all this simplification, we can rewrite (10.31) as

$$\vec{e}_k(\theta_{\vec{v}}) = \vec{c} + (\vec{e}_k - \vec{c}) \cos(\theta) + \frac{\vec{c} \times \vec{e}_k}{|\vec{c}|} \sin(\theta) \quad (10.37)$$

For  $k = 1, 2, 3$ , we can break the above expression into components:

$$\begin{aligned} \vec{e}_1(\theta_{\vec{v}}) &= \langle \alpha^2 + (1 - \alpha^2) \cos(\theta), \alpha\beta - \alpha\beta \cos(\theta) + \delta \sin(\theta), \alpha\delta - \alpha\delta \cos(\theta) - \beta \sin(\theta) \rangle \\ \vec{e}_2(\theta_{\vec{v}}) &= \langle \alpha\beta - \alpha\beta \cos(\theta) - \delta \sin(\theta), \beta^2 + (1 - \beta^2) \cos(\theta), \beta\delta - \beta\delta \cos(\theta) + \alpha \sin(\theta) \rangle \\ \vec{e}_3(\theta_{\vec{v}}) &= \langle \alpha\delta - \alpha\delta \cos(\theta) + \beta \sin(\theta), \beta\delta - \beta\delta \cos(\theta) - \alpha \sin(\theta), \delta^2 + (1 - \delta^2) \cos(\theta) \rangle \end{aligned}$$

So given a vector  $\vec{z} = \langle z_1, z_2, z_3 \rangle$ , we can express this as  $\vec{z} = z_1 \vec{e}_1 + z_2 \vec{e}_2 + z_3 \vec{e}_3$ . If we wish to rotate  $\vec{z}$  about a vector  $\vec{v}$  by angle  $\theta$ , the unknown matrix  $A_{\vec{v}}$  applied to  $\vec{z}$  satisfies:

$$\begin{aligned} A_{\vec{v}} \vec{z} &= z_1 A_{\vec{v}} \vec{e}_1 + z_2 A_{\vec{v}} \vec{e}_2 + z_3 A_{\vec{v}} \vec{e}_3 \\ &= z_1 \vec{e}_1(\theta_{\vec{v}}) + z_2 \vec{e}_2(\theta_{\vec{v}}) + z_3 \vec{e}_3(\theta_{\vec{v}}) \\ &= \left[ \vec{e}_1(\theta_{\vec{v}}) \mid \vec{e}_2(\theta_{\vec{v}}) \mid \vec{e}_3(\theta_{\vec{v}}) \right] \begin{bmatrix} z_1 \\ z_2 \\ z_3 \end{bmatrix} \end{aligned}$$

Therefore, now we have that the matrix  $A_{\vec{v}}$ 's columns are  $\vec{e}_1$ ,  $\vec{e}_2$ , and  $\vec{e}_3$ :

$$\begin{aligned} A_{\vec{v}} &= \quad (10.38) \\ &\begin{bmatrix} \alpha^2 + (1 - \alpha^2) \cos(\theta) & \alpha\beta - \alpha\beta \cos(\theta) + \delta \sin(\theta) & \alpha\delta - \alpha\delta \cos(\theta) - \beta \sin(\theta) \\ \alpha\beta - \alpha\beta \cos(\theta) - \delta \sin(\theta) & \beta^2 + (1 - \beta^2) \cos(\theta) & \beta\delta - \beta\delta \cos(\theta) + \alpha \sin(\theta) \\ \alpha\delta - \alpha\delta \cos(\theta) + \beta \sin(\theta) & \beta\delta - \beta\delta \cos(\theta) - \alpha \sin(\theta) & \delta^2 + (1 - \delta^2) \cos(\theta) \end{bmatrix} \end{aligned}$$



Comparing the matrix  $A_{\vec{v}}$  above to the matrix labeled *rotmatrix* in the previous set of *Maple* code reveals that we have arrived at the same rotation matrix as that found by *Maple*.

**Example 10.4.1.** After all of these calculations, it is time to do an example. Let  $\vec{w} = \langle -5, 3, 11 \rangle$  be a vector parallel to our axis of rotation that is a line through the origin. We want to rotate the point  $P(4, -7, 6)$  about this line through angles  $\theta$ , which are multiples of  $10^\circ$  until we are back at the point  $P$ . The plot of these points is essentially the circle of rotation for  $P$ , as seen in Figures 10.10 and 10.11.

```
> P:= [4,-7,6]: w:= [-5,3,11]:
> length_w:= norm(w,frobenius);
```

$$\text{length\_w} := \sqrt{155}$$

```
> v:= evalm(w/length_w):
> points:= [seq(evalm(evalf(A(op(convert(v,list)),2*Pi*j/36))&*P),j=0..36)]:
> points[1];
      [ 4.  -7.  6. ]
> plotpts:= seq(sphere(convert(points[j],list),.3),j=1..37):
> text_P:= textplot3d([op(P),“P”], align={ABOVE, RIGHT}, color=black):
> rotaxis:= line(convert(evalm(-10*v),list), convert(evalm(10*v),list), thick-
ness = 2, color = red):
> Center:= convert(evalm(dotprod(v,P)*v),list);
```

$$\text{Center} := \left[ -\frac{25}{31}, \frac{15}{31}, \frac{55}{31} \right]$$

```
> plot_Center:= sphere(Center,.3,style = patchnograd):
> text_center:= textplot3d([op(Center),“C”], align={ABOVE,RIGHT}, color
= black):
> Arrows:= [seq(PLOT3D(arrow(Center, convert(points[j],list), .15, .45, .25,
cylindrical_arrow, color = blue)), j = 1..37)]:
```

```
> display({rotaxis, text_P, text_center, plot_Center, display([plotpts], insequence = true), display(Arrows, insequence = true) }, axes = boxed, scaling = constrained);
```




Figure 10.10: First frame of a sphere located at  $P(4, -7, 6)$  being rotated about a line in the direction  $\vec{w} = \langle -5, 3, 11 \rangle$  in  $\mathbb{R}^3$ .

An alternative plot of this rotation is below, but it takes much more time and memory to compute since it plots simultaneously 37 spheres at each of the points at  $10^\circ$  of separation in the rotation of the point  $P$ .

```
> display({rotaxis, text_P, text_center, plot_Center, plotpts, display(Arrows, insequence = true) }, axes = boxed, scaling = constrained);
```




Figure 10.11: All of the spheres in the rotation about the line in the direction  $\vec{w} = \langle -5, 3, 11 \rangle$  are depicted here.

Now that we have determined all of the rotational matrices for  $\mathbb{R}^3$ , we will next focus our attention on rescalings. Similar to the two-dimensional case, a general three-dimensional rescaling is given by matrices of the form

$$A_{(a,b,c)} = \begin{bmatrix} a & 0 & 0 \\ 0 & b & 0 \\ 0 & 0 & c \end{bmatrix}$$

for real entries  $a$ ,  $b$ , and  $c$ . The above scaling matrix corresponds to the linear map  $T(\langle x, y, z \rangle) = \langle ax, by, cz \rangle$ . When  $a$ ,  $b$ , and  $c$  are all positive, we simply have a scaling in the same direction as  $\langle x, y, z \rangle$ . However, if at least one of the components along the diagonal is negative, we have not just a scaling, but a reflection as well. The matrix

$$A_{xy} = \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & -1 \end{bmatrix}$$

is reflection about the  $xy$ -plane, since the  $x$  and  $y$  coordinates stay fixed, and the  $z$  coordinate is replaced by  $-z$ . The matrix

$$A_{\vec{0}} = \begin{bmatrix} -1 & 0 & 0 \\ 0 & -1 & 0 \\ 0 & 0 & -1 \end{bmatrix}$$

is reflection about the origin. The question of whether all linear maps  $T : \mathbb{R}^3 \rightarrow \mathbb{R}^3$  can be decomposed into a composite of rotations and general rescalings is obviously a challenging one, and one should consider all of the work done in Section 10.2 to decompose all real  $2 \times 2$  matrices into a product of rotations, reflections and rescaling matrices.

## Homework Problems

1. Prove that the following relations hold, and give a geometric interpretation of this result

$$A_{\theta_x}^{-1} = A_{-\theta_x}, \quad A_{\theta_y}^{-1} = A_{-\theta_y}, \quad A_{\theta_z}^{-1} = A_{-\theta_z}$$

2. Verify that no two rotation matrices commute, (e.g.  $A_{\theta_x}A_{\phi_y} \neq A_{\phi_y}A_{\theta_x}$ ). Remember to use different angle variables for each matrix.

3. As a general rule, scalings and rotations do not commute. However, prove that scalings in just one direction commute with rotations corresponding to the axis of scaling for each of the  $x$ ,  $y$ , and  $z$  axes. As an example, show that

$A_{(a,1,1)}A_{\theta_x} = A_{\theta_x}A_{(a,1,1)}$ . Can you give a geometric interpretation of this result?

4. For rotations in the plane, it was shown that the rotation matrix  $A_\theta$  satisfied  $A_\theta A_\phi = A_{\theta+\phi}$ . Verify that the same property holds for the three rotation matrices  $A_{\theta_x}$ ,  $A_{\theta_y}$ , and  $A_{\theta_z}$  for rotations in  $\mathbb{R}^3$ .

5. Determine the  $3 \times 3$  matrix that will rotate the point  $P$  about the line passing through the origin in the direction of  $\vec{v}$  by an angle of  $\theta$ , then compute the coordinates of the rotated point.

(a)  $P(1, 1, 1)$ ,  $\vec{v} = \langle 3, 0, 0 \rangle$ ,  $\theta = \frac{\pi}{4}$       (b)  $P(1, 0, -1)$ ,  $\vec{v} = \langle 0, 2, 0 \rangle$ ,  $\theta = \frac{\pi}{2}$

(c)  $P(1, 0, 1)$ ,  $\vec{v} = \langle 0, 0, 1 \rangle$ ,  $\theta = \frac{3}{4}\pi$       (d)  $P(0, 1, 1)$ ,  $\vec{v} = \langle 1, 0, 0 \rangle$ ,  $\theta = \frac{3}{2}\pi$

6. For each of the following, construct a single matrix that will satisfy each of the following set of criteria, in the sequence specified.

(a) First rotate about the  $z$ -axis by  $\theta = \frac{\pi}{3}$ , then scale in the  $x$ -direction by a factor of 2, and finally, scale in the  $y$ -direction by a factor of 3.

(b) First flip about the  $xy$ -plane, next rotate about  $z$ -axis by  $\theta = \frac{\pi}{4}$ , and lastly, scale in  $z$ -direction by a factor of 2.

(c) First scale in  $y$ -direction by a factor of 2, then rotate about  $y$ -axis by an angle of  $\theta = -\frac{\pi}{2}$ , then finally, scale in the  $x$ -direction by a factor of  $\frac{1}{2}$ .

7. Describe a procedure for rotating a point  $P(x, y, z)$  about the line passing through the origin in the direction of  $\vec{v} = \langle \alpha, \beta, 0 \rangle$ , where  $\alpha$  and  $\beta$  are both non-zero and satisfy  $\alpha^2 + \beta^2 = 1$ .

8. Decompose an arbitrary  $3 \times 3$  matrix into the product of rotations, rescalings and reflections.

## Maple Problems

1. *Homework* problem 1 illustrated the fact that the inverse rotation matrices can be found by replacing the angle  $\theta$  with  $-\theta$  in the original rotation matrix. We wish to determine if the the same holds true for the rotation matrix  $A_{\vec{v}}$ .

- (a) Replace  $\theta$  with  $-\theta$  in  $A_{\vec{v}}$  and perform the matrix multiplications

$$A_{\vec{v}}(\theta)A_{\vec{v}}(-\theta) \text{ and } A_{\vec{v}}(-\theta)A_{\vec{v}}(\theta)$$

- (b) Compute  $A_{\vec{v}}^{-1}$  using *Maple's* inverse command, simplify and compare the result to  $A_{\vec{v}}(-\theta)$ .

2. Determine the  $3 \times 3$  matrix that will rotate the point  $P$  about the line passing through the origin in the direction of  $\vec{v}$  by an angle of  $\theta$ .

(a)  $P(2, 0, 3)$ ,  $\vec{v} = \langle 2, -1, 2 \rangle$       (b)  $P(2, 3, -1)$ ,  $\vec{v} = \langle -1, 0, 0 \rangle$

(c)  $P(1, 0, 1)$ ,  $\vec{v} = \langle 0, 0, 1 \rangle$       (d)  $P(1, -1, 0)$ ,  $\vec{v} = \langle 1, 2, 1 \rangle$

(e)  $P(0, 1, 1)$ ,  $\vec{v} = \langle 1, 0, 0 \rangle$       (f)  $P(1, -1, 3)$ ,  $\vec{v} = \langle -1, 1, 2 \rangle$

3. Plot, and animate, an entire rotation of the point  $P$  about the line in the direction  $\vec{v}$  for each of parts (a)-(f) of problem 2.

4. Describe and test a procedure for doing rotations about any line in space, not necessarily through the origin.

## 10.5 Affine Maps

Closely related to linear maps are affine maps. We begin with a definition:

**Definition 10.5.1.** An *affine map*  $S : \mathbb{R}^n \rightarrow \mathbb{R}^m$  is a translation of a linear map, that is,  $S(\vec{u}) = T(\vec{u}) + \vec{b}$  for all  $u \in \mathbb{R}^n$  where  $T : \mathbb{R}^n \rightarrow \mathbb{R}^m$  is a linear map and  $\vec{b} \in \mathbb{R}^m$  is fixed.

If  $A$  is the standard  $m \times n$  matrix representing the linear map  $T$ , then  $A$ , along with the fixed vector  $\vec{b}$  in  $\mathbb{R}^m$ , also represents the affine map  $S$ , since we have  $S(\vec{u}) = A\vec{u} + \vec{b}$  for all  $\vec{u} \in \mathbb{R}^n$ . Affine maps operate analogously to linear maps, except that they also translate their results by  $\vec{b}$ . Thus, affine maps will take a subspace  $\mathbb{K} \subseteq \mathbb{R}^n$  and return  $S(\mathbb{K}) = \mathbb{L} + \vec{b}$  in  $\mathbb{R}^m$ , where  $\mathbb{L} = T(\mathbb{K})$  is the subspace of  $\mathbb{R}^m$  corresponding to the image of the subspace  $\mathbb{K}$  of  $\mathbb{R}^n$  under the map  $T$ . Hence, an affine map  $S$  translates the image subspace of  $T$  by  $\vec{b}$ . This should remind you of the work we did back in Section 8.1 on subspace translates. The fixed vector  $\vec{b}$  of  $\mathbb{R}^m$  is  $S(\vec{0}_n)$ , since

$$S(\vec{0}_n) = T(\vec{0}_n) + \vec{b} = \vec{0}_m + \vec{b} = \vec{b}$$

Note that we used the fact that linear maps take the zero vector of their domain to the zero vector of their range. We can directly relate solving a linear system of equations  $A\vec{x} = -\vec{b}$ , for an  $m \times n$  matrix  $A$ , to affine maps. This matrix equation can be written as  $A\vec{x} + \vec{b} = \vec{0}_m$ . The solutions to this matrix equation are the solutions  $\vec{u}$  in  $\mathbb{R}^n$  to  $S(\vec{u}) = \vec{0}_m$ , which is to say, solutions to  $T(\vec{u}) + \vec{b} = \vec{0}_m$ , for the linear map  $T$  with standard matrix  $A$  representing it. We already know that the solutions  $\vec{x}$  in  $\mathbb{R}^n$  to  $A\vec{x} = -\vec{b}$  are of the form  $\vec{x} = \vec{x}_k + \vec{c}$ , where  $\vec{x}_k \in \text{Ker}(T)$  corresponds to the solutions of  $A\vec{x} = \vec{0}_m$ , and  $\vec{c}$  is a fixed vector in  $\mathbb{R}^n$ , with  $A\vec{c} = -\vec{b}$ .

**Example 10.5.1.** We next do an example of an affine map. Consider  $S : \mathbb{R}^3 \rightarrow \mathbb{R}^4$  given by  $S(\vec{u}) = A\vec{u} + \vec{b}$ , where

$$A = \begin{bmatrix} -3 & 1 & 7 \\ 1 & 9 & 3 \\ -1 & 19 & 13 \\ 4 & 8 & -4 \end{bmatrix}, \quad \vec{b} = \begin{bmatrix} -8 \\ 5 \\ 2 \\ 13 \end{bmatrix}$$

In the following *Maple* commands, we plot (Figure 10.12) the solutions to  $S(\vec{u}) = \vec{0}_4$ . In other words, we are searching for solutions to  $A\vec{u} = -\vec{b}$ .

```
> with(linalg): with(plots): with(plottools):
> A:= matrix([[ -3, 1, 7], [1, 9, 3], [-1, 19, 13], [4, 8, -4]]):
> B:= matrix([[8], [-5], [-2], [-13]]):
> rref(augment(A,B)):
```

$$\begin{bmatrix} 1 & 0 & -\frac{15}{7} & -\frac{11}{4} \\ 0 & 1 & \frac{4}{7} & -\frac{1}{4} \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

From the solution given above, we can reconstruct the solution as a line given parametrically as

$$L = \left\{ \langle x, y, z \rangle \mid x = \frac{15}{7}t - \frac{11}{4}, y = -\frac{4}{7}t - \frac{1}{4}, z = t \right\} \quad (10.39)$$

for  $t$  an arbitrary parameter. We can decompose  $L$  into a sum of two components, the first being the line  $L_0$ , which passes through the origin, the second being a translation by a particular vector  $\vec{x}_p$ . To find the particular vector, notice that all we have to do is set  $t = 0$  in the parametric definition of  $L$  given

above, which yields  $\vec{x}_p = \langle -\frac{11}{4}, -\frac{1}{4}, 0 \rangle$ . Now that we have  $\vec{x}_p$ , the line  $L_0$  is simply the remaining portion of the solution given in equation 10.39:

$$L_0 = \left\{ \left\langle \frac{15}{7}t, -\frac{4}{7}t, t \right\rangle \mid t \in \mathbb{R} \right\} \quad (10.40)$$

Clearly  $L_0$  is a line through the origin, and is thus a subspace of  $\mathbb{R}^3$ . As previously discussed, the line  $L$  can be realized as a translate of the line  $L_0$  by the particular solution  $\vec{x}_p$ . Now, let us plot the two lines  $L$  and  $L_0$  along with the particular solution vector  $\vec{x}_p$ .

```
> plotL:= spacecurve([15/7*t-11/4, -4/7*t-1/4, t], t = -3..3, color=red, thick-
ness=2):
> plotL0:= spacecurve([15/7*t, -4/7*t, t], t = -3..3, color = blue, thickness
= 2):
> plotP:= arrow([0,0,0], [-11/4,-1/4,0], .1, .2, .5, cylindrical_arrow, color =
black):
> plot_origin:= textplot3d([0,0,0,"O"], align={LEFT, ABOVE}, color =
black):
> plot_P:= textplot3d([-11/4,-1/4,0,"P"], align={LEFT, ABOVE}, color =
black):
> display({plotL, plotL0, plotP, plot_origin, plot_P}, axes = boxed);
```




Figure 10.12: The line  $L$  can be decomposed into a line through the origin and a particular vector solution.

---

Recall that a linear map  $T : \mathbb{R}^2 \rightarrow \mathbb{R}^2$  is uniquely determined by taking a line segment in the domain to another line segment in the range. This is no longer the case for an affine map  $S : \mathbb{R}^2 \rightarrow \mathbb{R}^2$ . As it turns out, an affine map  $S : \mathbb{R}^2 \rightarrow \mathbb{R}^2$  is uniquely determined by taking a triangle in the domain to

another triangle in the range. To see how this is the case, let the triangle in the domain be defined as the interior of the three points that are the terminal points of the following three vectors:

$$T_1 = \{\langle x_1, y_1 \rangle, \langle x_2, y_2 \rangle, \langle x_3, y_3 \rangle\}$$

similarly for the triangle in the range:

$$T_2 = \{\langle z_1, w_1 \rangle, \langle z_2, w_2 \rangle, \langle z_3, w_3 \rangle\}$$

Then,

$$S(\langle x_j, y_j \rangle) = \langle z_j, w_j \rangle, 1 \leq j \leq 3$$

If  $S(\vec{u}) = A\vec{u} + \vec{b}$ , for the  $2 \times 2$  matrix  $A = \begin{bmatrix} a & b \\ c & d \end{bmatrix}$  and  $\vec{b} = \begin{bmatrix} \alpha \\ \beta \end{bmatrix}$ , then we have the systems of equations

$$\begin{bmatrix} a & b \\ c & d \end{bmatrix} \begin{bmatrix} x_j \\ y_j \end{bmatrix} + \begin{bmatrix} \alpha \\ \beta \end{bmatrix} = \begin{bmatrix} z_j \\ w_j \end{bmatrix}, \quad 1 \leq j \leq 3$$

This gives the new single matrix equation

$$\begin{bmatrix} x_1 & y_1 & 0 & 0 & 1 & 0 \\ 0 & 0 & x_1 & y_1 & 0 & 1 \\ x_2 & y_2 & 0 & 0 & 1 & 0 \\ 0 & 0 & x_2 & y_2 & 0 & 1 \\ x_3 & y_3 & 0 & 0 & 1 & 0 \\ 0 & 0 & x_3 & y_3 & 0 & 1 \end{bmatrix} \begin{bmatrix} a \\ b \\ c \\ d \\ \alpha \\ \beta \end{bmatrix} = \begin{bmatrix} z_1 \\ w_1 \\ z_2 \\ w_2 \\ z_3 \\ w_3 \end{bmatrix} \quad (10.41)$$

This last matrix equation can be solved for our six unknowns  $\{a, b, c, d, \alpha, \beta\}$ , which determine the affine map uniquely.

---

**Example 10.5.2.** As an example, we will now find the affine map  $S$  that sends the triangle  $T_1$  to triangle  $T_2$  defined by the following sets of points:

$$T_1 = \{\langle -9, -5 \rangle, \langle 4, 17 \rangle, \langle 1, -6 \rangle\}$$

$$T_2 = \{\langle 3, 11 \rangle, \langle -8, 3 \rangle, \langle -2, -15 \rangle\}$$

> X:= [-9,4,1]: Y:= [-5,17,-6]: Z:= [3,-8,-2]: W:= [11,3,-15]:

> R:= matrix([[Z[1]], [W[1]], [Z[2]], [W[2]], [Z[3]], [W[3]]]);

$$R := \begin{bmatrix} 3 \\ 11 \\ -8 \\ 3 \\ -2 \\ -15 \end{bmatrix}$$



```
> d:= matrix([[X[1], Y[1], 0, 0, 1, 0], [0, 0, X[1], Y[1], 0, 1], [X[2], Y[2], 0, 0, 1, 0], [0, 0, X[2], Y[2], 0, 1], [X[3], Y[3], 0, 0, 1, 0], [0, 0, X[3], Y[3], 0, 1]]);
```

$$\begin{bmatrix} -9 & -5 & 0 & 0 & 1 & 0 \\ 0 & 0 & -9 & -5 & 0 & 1 \\ 4 & 17 & 0 & 0 & 1 & 0 \\ 0 & 0 & 4 & 17 & 0 & 1 \\ 1 & -6 & 0 & 0 & 1 & 0 \\ 0 & 0 & 1 & -6 & 0 & 1 \end{bmatrix}$$

```
> AB:= evalm(inverse(d)&*R);
```

$$AB := \begin{bmatrix} -\frac{121}{233} \\ -\frac{45}{233} \\ -\frac{580}{233} \\ \frac{258}{233} \\ -\frac{615}{233} \\ -\frac{1367}{233} \end{bmatrix}$$

```
> A:= matrix([[AB[1,1],AB[2,1]], [AB[3,1],AB[4,1]]]);
```

$$A := \begin{bmatrix} -\frac{121}{233} & -\frac{45}{233} \\ -\frac{580}{233} & \frac{258}{233} \end{bmatrix}$$

```
> B:= matrix([[AB[5,1]], [AB[6,1]]]);
```

$$B := \begin{bmatrix} -\frac{615}{233} \\ -\frac{1367}{233} \end{bmatrix}$$

```
> evalm(A&*[X[1],Y[1]]+B);
```

$$\begin{bmatrix} 3 \\ 11 \end{bmatrix}$$

```
> evalm(A&*[X[2],Y[2]]+B);
```

$$\begin{bmatrix} -8 \\ 3 \end{bmatrix}$$

```
> evalm(A&*[X[3],Y[3]]+B);
```

$$\begin{bmatrix} -2 \\ -15 \end{bmatrix}$$

The last three *Maple* commands are simply verifications that the vectors  $\langle x_k, y_k \rangle$ , determining the corners of triangle  $T_1$ , were sent to their corresponding

counterparts  $\langle z_k, w_k \rangle$ , of  $T_2$ . Hence, our formula works. Now you should see what it takes to determine an affine map  $S: \mathbb{R}^3 \rightarrow \mathbb{R}^3$  uniquely, and in general when the dimensions are not the same for both domain and range.

We have previously looked at rotations about the origin in  $\mathbb{R}^2$  and rotations about a line through the origin in  $\mathbb{R}^3$ . The next logical question to ask, is what happens if we rotate about a point  $(a, b)$  in  $\mathbb{R}^2$  which is not the origin, or rotate about a line that does not pass through the origin in  $\mathbb{R}^3$ ? You should have guessed that we get an affine map for both rotations. In Section 8.1, we already expressed this concept in terms of matrix multiplication, so it is fairly straightforward to rewrite the ideas in terms of maps instead.

We will once again define  $A_\theta$  to be the  $2 \times 2$  rotation matrix for the fixed angle  $\theta$  about the origin in  $\mathbb{R}^2$ , defined in the standard way. If we now wish to rotate a point corresponding to the terminal end of a vector  $\vec{u}$  about the point  $(a, b)$ , we need to use the affine map  $S$ . Here,  $S$  is defined as follows:

$$\begin{aligned} S(\vec{u}) &= A_\theta(\vec{u} - \langle a, b \rangle) + \langle a, b \rangle \\ &= A_\theta \vec{u} + \langle a, b \rangle - A_\theta \langle a, b \rangle \\ &= A_\theta \vec{u} + \vec{b} \end{aligned} \tag{10.42}$$

where

$$\begin{aligned} \vec{b} &= \langle a, b \rangle - A_\theta \langle a, b \rangle \\ &= (I_2 - A_\theta) \langle a, b \rangle \end{aligned} \tag{10.43}$$

Notice this is similar to the matrix multiplication version from Section ??, where we translated our frame of rotation from the point  $(a, b)$  to the origin, and then translated back by adding the point  $(a, b)$  to the resulting rotated point. The same can be done for a rotation about a line  $L$  not through the origin in  $\mathbb{R}^3$ . If  $A_\theta$  is the  $3 \times 3$  rotation matrix for the fixed angle  $\theta$  about the line  $L'$  parallel to  $L$  but through the origin, then the affine map  $S$  will rotate about  $L$  any point whose represented by the terminal end of a vector  $\vec{u} \in \mathbb{R}^3$ , and is defined as

$$\begin{aligned} S(\vec{u}) &= A_\theta(\vec{u} - \langle a, b, c \rangle) + \langle a, b, c \rangle \\ &= A_\theta \vec{u} + \langle a, b, c \rangle - A_\theta \langle a, b, c \rangle = A_\theta \vec{u} + \vec{b} \end{aligned} \tag{10.44}$$

for

$$\begin{aligned} \vec{b} &= \langle a, b, c \rangle - A_\theta \langle a, b, c \rangle \\ &= (I_3 - A_\theta) \langle a, b, c \rangle \end{aligned} \tag{10.45}$$

Here the point  $(a, b, c)$  is a fixed point on the line  $L$ .

**Example 10.5.3.** Now, let us do an example of this three-dimensional rotation about a line  $L$  that does not pass through the origin. First, we consider the line going through the point  $\vec{P}\langle 11, 25, -7\rangle$ , which is parallel to the vector  $\vec{w} = \langle 5, 2, 9\rangle$ . We want to rotate the point  $\vec{Q}\langle 4, -7, 13\rangle$  about this line through angles  $\theta$  that are multiples of  $10^\circ$ , until we are back at the point  $\vec{Q}$ . Each rotation through a fixed angle  $\theta$  is one application of an affine map. The plot of these points, which can be seen in Figures 10.13 and 10.14, is essentially the circle of rotation for  $\vec{Q}$ . Referring back to Section 4.4, the formula for the rotated point  $\vec{Q}_{\text{new}}$  will be given by

$$\begin{aligned}\vec{Q}_{\text{new}} &= A_{\vec{v}}(\vec{Q} - \vec{P}) + \vec{P} \\ &= (I_3 - A_{\vec{v}})\vec{P} + A_{\vec{v}}\vec{Q}\end{aligned}\tag{10.46}$$

In the *Maple* commands below, the lengthy formula defining  $A$  below is simply the expression for  $A_{\vec{v}}$  given in equation (10.38), and the matrix  $\vec{B}$  corresponds to the column matrix  $(I_3 - A_{\vec{v}})\vec{P}$ , giving the formula for  $\vec{Q}_{\text{new}}$  to now be

$$\vec{Q}_{\text{new}} = A_{\vec{v}}\vec{Q} + \vec{B}$$

```
> A:= (alpha,beta,delta,theta) -> matrix([[alpha^2 + cos(theta) - cos(theta)
*alpha^2, alpha*beta - cos(theta)*alpha*beta + sin(theta)*delta, alpha*delta
- cos(theta)*alpha*delta - sin(theta)*beta], [alpha*beta - cos(theta)*alpha
*beta - sin(theta)*delta, beta^2 + cos(theta) - cos(theta)*beta^2, beta*delta -
cos(theta)*beta*delta + sin(theta)*alpha], [alpha*delta - cos(theta)*alpha
*delta + sin(theta)*beta, beta*delta - cos(theta)*beta*delta - sin(theta)*alpha,
delta^2 + cos(theta) - cos(theta)*delta^2]]);
> Q:= [4,-7,13]: P:= [11, 25, -7]: w:= [5, 2, 9]:
> length_w:= norm(w,frobenius);
```

$$\text{length\_w} := \sqrt{110}$$

```
> v:= evalm(w/length_w);
```

$$v := \begin{bmatrix} \frac{\sqrt{110}}{22}, \frac{\sqrt{110}}{55}, \frac{9\sqrt{110}}{110} \end{bmatrix}$$

```
> B:= theta -> evalm((diag(1,1,1)-evalf(A(op(convert(v,list)), theta)))*P):
> points:= [seq(evalm(evalf(A(op(convert(v,list)), 2*Pi*j / 36)) &*Q + B(
2*Pi*j / 36)), j = 0..36)];
> points[1];
```

$$\begin{bmatrix} 4. & -7. & 13. \end{bmatrix}$$

```

> plotpts:= seq(sphere(convert(points[j], list), 1), j=1..37):
> text_Q:= textplot3d([op(Q), "Q"], align={ABOVE,RIGHT}, color=black):
> rotaxis:= line(convert(evalm(-30*v + P), list), convert(evalm(30*v + P),
list), thickness = 2, color = red):
> Center:= convert(evalm(dotprod(v,Q)*v+P),list);

```

$$Center := \left[ \frac{365}{22}, \frac{1498}{55}, \frac{337}{110} \right]$$

```

> plot_Center:= sphere(Center,1):
> text_center:= textplot3d([op(Center), "C"], align={ABOVE, RIGHT}, color
= black):
> plot_P:= sphere(P,1):
> text_P:= textplot3d([op(P), "P"],align={BELOW, RIGHT},color=black):
> plot_origin:= sphere([0,0,0],1):
> text_origin:= textplot3d([op([0,0,0]), "O"], align={ABOVE, RIGHT}, color
= black):
> Arrows:= [seq(PLOT3D(arrow(Center, convert(points[j], list), 1,5,.3, cylin-
drical_arrow, color = blue)), j=1..37)]:
> display({rotaxis, text_Q, plot_P, text_P, text_center, plot_Center, plot_origin,
text_origin, display([plotpts], insequence = true), display(Arrows, insequence
= true)}, axes = boxed, scaling = constrained, view=[-40..60,-20..65,-30..30]);

```




Figure 10.13: The point  $\vec{Q}$  (at the tip of the arrow) is rotated about the line passing through the point  $\vec{P}$  (the point on the line not at the tail of the vector) in the direction of the vector  $\vec{w}$ . The origin is also shown for reference and is located directly below the point  $\vec{Q}$  in this figure.

```

> display({rotaxis, text_Q, plot_P, text_P, text_center, plot_Center, plot_origin,
text_origin, display([plotpts], insequence = false), display(Arrows[1])}, axes =

```

frame, scaling = constrained, view = [-40..60, -20..65, -30..30], labels = [x, y, z], orientation = [126, 64], style = patchnograd);




Figure 10.14: All of the rotated points, along with corresponding perpendicular arrows to the line of axis of rotation.

We end this section with a procedure called *RotationsInSpace* which takes a non-zero vector  $\vec{v} = \langle \alpha, \beta, \delta \rangle$  parallel to the line  $L$  of rotation where the line of rotation goes through the point  $\vec{P}$  (which could be the origin) but  $L$  is not parallel to any axis or coordinate plane (no component of  $\vec{v}$  can be 0), and then rotates a point  $\vec{Q}$  through the angle  $\theta$  about  $L$ . The vector  $\vec{v}$  and two points  $\vec{P}$  and  $\vec{Q}$  can be given as lists, although  $\vec{v}$  as a vector also works.

```
> with(linalg):
> RotationsInSpace:= proc(v,P,Q,theta)
  local alpha, beta, delta, A, B, w:
  w:= evalm(v/norm(v, frobenius)):
  alpha:= evalf(w[1]);
  beta:= evalf(w[2]);
  delta:= evalf(w[3]);
  A:= (alpha,beta,delta,theta) -> matrix([[alpha^2 + cos(theta) - cos(theta)
    *alpha^2, alpha*beta - cos(theta)*alpha*beta + sin(theta)*delta, alpha
    *delta - cos(theta)*alpha*delta - sin(theta)*beta], [alpha*beta - cos(theta)
    *alpha*beta - sin(theta)*delta,beta^2 + cos(theta) - cos(theta)*beta^2,
    beta*delta -cos(theta)*beta*delta + sin(theta)*alpha], [alpha*delta
    -cos(theta)*alpha*delta+ sin(theta)*beta, beta*delta - cos(theta)*beta
    *delta -sin(theta)*alpha, delta^2 + cos(theta) - cos(theta)*delta^2]):
  B:= evalm(P - A&*P):
  evalm(A&*Q + B):
end proc:
```

> Q1 := RotationsInSpace([5, 2, 9], [11, 25, -7], [4, -7, 13], Pi/3);

$$Q1 := [9.340909088 + 15.63678646 \sqrt{3}, 9.736363627 - 7.770720100 \sqrt{3}, \\ -6.960276898 \sqrt{3} + 6.313636364]$$

> v:= [5, 2, 9];

> w:= evalm(v/norm(v, frobenius));

$$w := \left[ \frac{1}{22} \sqrt{110} \quad \frac{1}{55} \sqrt{110} \quad \frac{9}{110} \sqrt{110} \right]$$

> P := [11, 25, -7]: Q := [4, -7, 13]:

> C := evalm(dotprod(w, Q)\*w+P);

$$C := \begin{bmatrix} \frac{365}{22} & \frac{1498}{55} & \frac{337}{110} \end{bmatrix}$$

> evalf(norm(Q-C, frobenius)); evalf(norm(Q1-C, frobenius));

37.80728645

37.80728645

## Homework Problems

1. Find the solutions to the equations  $S(\vec{u}) = \vec{0}$ , for each of the following affine maps:

(a)  $\begin{bmatrix} 1 & -2 \\ 3 & -4 \end{bmatrix} \vec{u} + \begin{bmatrix} -1 \\ 1 \end{bmatrix}$

(b)  $\begin{bmatrix} 4 & 3 \\ -1 & 2 \\ -2 & 5 \end{bmatrix} \vec{u} + \begin{bmatrix} -1 \\ 2 \\ 1 \end{bmatrix}$

(c)  $\begin{bmatrix} 0 & 2 \\ 5 & 0 \\ -5 & 2 \end{bmatrix} \vec{u} + \begin{bmatrix} 7 \\ -9 \\ 16 \end{bmatrix}$

(d)  $\begin{bmatrix} 1 & -2 & 2 \\ 0 & 3 & 2 \\ -2 & -4 & -1 \end{bmatrix} \vec{u} + \begin{bmatrix} -1 \\ 1 \\ 4 \end{bmatrix}$

(e)  $\begin{bmatrix} 1 & -2 & 2 \\ 0 & 3 & 2 \\ -2 & -4 & -1 \\ 4 & 5 & 6 \end{bmatrix} \vec{u} + \begin{bmatrix} -1 \\ 2 \\ -1 \\ 3 \end{bmatrix}$

(f)  $\begin{bmatrix} 3 & 5 & 1 \\ -2 & 1 & -2 \end{bmatrix} \vec{u} + \begin{bmatrix} -1 \\ 1 \end{bmatrix}$

2. State the dimension for each of the solutions to the affine maps from problem 1.

3. How many points in  $\mathbb{R}^n$  does it take to define an affine transformation  $S : \mathbb{R}^n \rightarrow \mathbb{R}^n$  uniquely?

4. Referring to the matrix equation (10.41), construct the matrix used to determine the number of vectors needed to create a unique affine map from  $\mathbb{R}^2$  to  $\mathbb{R}^3$  of the form:

$$S(\vec{u}) = \begin{bmatrix} a & b \\ c & d \\ e & f \end{bmatrix} \vec{u} + \begin{bmatrix} \alpha \\ \beta \\ \gamma \end{bmatrix}$$

5. Determine the number of vectors needed in  $\mathbb{R}^n$  and  $\mathbb{R}^m$ , with  $n, m \geq 1$ , which will determine an affine map  $S : \mathbb{R}^n \rightarrow \mathbb{R}^m$  uniquely. It may help to construct a generic matrix, as in problem 4 and equation (10.41).

6. Express, as an affine map, the rotation of an arbitrary vector  $\vec{u} \in \mathbb{R}^2$  through an angle  $\theta$  about the point  $(1, -1)$ .

7. Can a single affine transformation  $S : \mathbb{R}^2 \rightarrow \mathbb{R}^2$  send the interior and sides of the unit square to the interior and sides of an arbitrary quadrilateral in  $\mathbb{R}^2$ ? If yes, explain why. If no, can it be done for any special type of quadrilateral?

## Maple Problems

1. Construct an affine map  $S$  that takes the first set of points  $\mathbf{S}$  to the second set of points  $\mathbf{T}$ , with  $S(\vec{s}_k) = \vec{t}_k$ .

$$\begin{aligned} \text{(a) } \mathbf{S} &= \left\{ \vec{s}_1 = \langle 1, -1, 1 \rangle, \vec{s}_2 = \langle 0, 2, -3 \rangle, \vec{s}_3 = \langle 2, 4, -3 \rangle, \vec{s}_4 = \langle 10, -3, 2 \rangle \right\} \\ \mathbf{T} &= \left\{ \vec{t}_1 = \langle 1, 2 \rangle, \vec{t}_2 = \langle 4, 2 \rangle, \vec{t}_3 = \langle 4, 5 \rangle, \vec{t}_4 = \langle 9, -7 \rangle \right\} \end{aligned}$$

$$\begin{aligned} \text{(b) } \mathbf{S} &= \left\{ \vec{s}_1 = \langle 1, 2 \rangle, \vec{s}_2 = \langle -2, 3 \rangle, \vec{s}_3 = \langle 5, -1 \rangle \right\} \\ \mathbf{T} &= \left\{ \vec{t}_1 = \langle 1, 2, -1 \rangle, \vec{t}_2 = \langle 7, 8, -2 \rangle, \vec{t}_3 = \langle 3, -5, 6 \rangle \right\} \end{aligned}$$

$$\begin{aligned} \text{(c) } \mathbf{S} &= \left\{ \vec{s}_1 = \langle 1, 1, 3 \rangle, \vec{s}_2 = \langle -2, 0, 5 \rangle, \vec{s}_3 = \langle 4, 3, -3 \rangle, \vec{s}_4 = \langle 7, -6, 8 \rangle \right\} \\ \mathbf{T} &= \left\{ \vec{t}_1 = \langle 1, 2, -2, 1 \rangle, \vec{t}_2 = \langle 3, 5, 6, 7 \rangle, \vec{t}_3 = \langle -4, -6, 9, 0 \rangle, \vec{t}_4 = \langle -3, -2, 5, -3 \rangle \right\} \end{aligned}$$

2. Construct a family of affine maps that takes the first set of points  $\mathbf{S}$  to the second set of points  $\mathbf{T}$ .

$$\begin{aligned} \text{(a) } \mathbf{S} &= \left\{ \vec{s}_1 = \langle 1, -2 \rangle, \vec{s}_2 = \langle 9, 5 \rangle \right\} \\ \mathbf{T} &= \left\{ \vec{t}_1 = \langle 3, -8, -9, 1 \rangle, \vec{t}_2 = \langle 4, 5, 7, -6 \rangle \right\} \end{aligned}$$

$$\begin{aligned} \text{(b) } \mathbf{S} &= \left\{ \vec{s}_1 = \langle -4, 5, 0, 3 \rangle, \vec{s}_2 = \langle 0, 3, -2, 1 \rangle, \vec{s}_3 = \langle 2, 5, 8, -1 \rangle \right\} \\ \mathbf{T} &= \left\{ \vec{t}_1 = \langle 1, 6, 3 \rangle, \vec{t}_2 = \langle -4, 3, 7 \rangle, \vec{t}_3 = \langle 2, 3, -2 \rangle \right\} \end{aligned}$$

3. Rotate the point  $(-3, 4)$  about the point  $(1, 1)$ . Perform a complete rotation, including at least ten frames in your animation.
4. Rotate the point  $(-10, -9, 9)$  about the line that passes through the point  $(1, 1, 1)$  and points in the direction  $\langle 2, -2, 4 \rangle$ . Perform a complete rotation, including at least ten frames in your animation.
5. Rotate a sphere of radius 2, with center  $(3, 3, 0)$ , about the line that passes through the point  $(-1, -1, 1)$ , and points in the direction perpendicular to the line that passes through the center of the sphere and the point  $(-1, -1, 1)$ . Perform a complete rotation, including at least ten frames in your animation.

## Research Projects

1. An interesting application of affine maps has been used by Michael Barnsley to construct geometric objects called fractals. Research the topic of fractals in Barnsley's book *Fractals Everywhere* and write *Maple* code to generate his Fern fractal example. You might also look at the book *Elementary Linear Algebra with Applications*, 7th edition (or later) by Howard Anton and Chris Rorres for some information on fractals.